

## Essential Maple 7

Robert M. Corless  
Department of Applied Mathematics  
University of Western Ontario  
London, Canada

### Copyright

Copyright 2001 by Robert M. Corless  
All rights reserved

### Chapter 1: Basics

#### Section 1.1: Getting Started

##### *Session 1.1.3: Sample Maple sessions*

##### *First sample session - Maple as a calculator*

```
> restart;  
  
> FactoredForm := factor( t^12 - 1 );
```

$$\text{FactoredForm} := (t - 1) (t^2 + t + 1) (t + 1) (1 - t + t^2) (t^2 + 1) (t^4 - t^2 + 1)$$

```
> expand( FactoredForm );
```

$$t^{12} - 1$$

```
> restart;  
  
> with(LinearAlgebra);  
  
> A := Matrix([[4,5], [5,6]]);
```

$$A := \begin{bmatrix} 4 & 5 \\ 5 & 6 \end{bmatrix}$$

> **A.A;**

$$\begin{bmatrix} 41 & 50 \\ 50 & 61 \end{bmatrix}$$

> **CharacteristicPolynomial( A, lambda );**

$$-1 - 10\lambda + \lambda^2$$

> **restart;**

> **intEqn := Int( 1/(t^6-1), t ) = int( 1/(t^6-1), t );**

$$\text{intEqn} := \int \frac{1}{t^6 - 1} dt = \frac{1}{6} \ln(t - 1) - \frac{1}{12} \ln(t^2 + t + 1) - \frac{1}{6} \sqrt{3} \arctan\left(\frac{1}{3}(2t + 1)\sqrt{3}\right) - \frac{1}{6} \ln(t + 1) + \frac{1}{12} \ln(t^2 - t + 1) - \frac{1}{6} \sqrt{3} \arctan\left(\frac{1}{3}(2t - 1)\sqrt{3}\right)$$

> **backAgain := diff( intEqn, t );**

$$\text{backAgain} := \frac{1}{t^6 - 1} = \frac{1}{6} \frac{1}{t - 1} - \frac{1}{12} \frac{2t + 1}{t^2 + t + 1} - \frac{1}{3} \frac{1}{1 + \frac{1}{3}(2t + 1)^2} - \frac{1}{6} \frac{1}{t + 1} + \frac{1}{12} \frac{(2t - 1)}{t^2 - t + 1} - \frac{1}{3} \frac{1}{1 + \frac{1}{3}(2t - 1)^2}$$

> **normal( backAgain, expanded );**

$$\frac{1}{t^6 - 1} = \frac{1}{t^6 - 1}$$

> **restart;**

> **Logistic := diff(x(t), t) = x(t)\*(1-x(t));**

$$\text{Logistic} := \frac{\partial}{\partial t} x(t) = x(t) (1 - x(t))$$

> **initialCond := x(0) = alpha;**

$$\text{initialCond} := x(0) = \alpha$$

> **ans := dsolve( {Logistic,initialCond}, x(t) );**

$$\text{ans} := x(t) = \frac{1}{1 - \frac{e^{(-t)} (-1 + \alpha)}{\alpha}}$$

> **check := eval( Logistic, ans );**

$$\text{check} := - \frac{e^{(-t)} (-1 + \alpha)}{\left(1 - \frac{e^{(-t)} (-1 + \alpha)}{\alpha}\right)^2 \alpha} = \frac{1 - \frac{1}{1 - \frac{e^{(-t)} (-1 + \alpha)}{\alpha}}}{1 - \frac{e^{(-t)} (-1 + \alpha)}{\alpha}}$$

> **normal( lhs(check) - rhs(check) );**

0

> **check := subs( ans, Logistic );**

$$check := \frac{\partial}{\partial t} \frac{1}{1 - \frac{e^{(-t)}(-1 + \alpha)}{\alpha}} = \frac{1 - \frac{1}{1 - \frac{e^{(-t)}(-1 + \alpha)}{\alpha}}}{1 - \frac{e^{(-t)}(-1 + \alpha)}{\alpha}}$$

> **normal( (lhs-rhs)(check) );**

0

> **odetest(ans, Logistic);**

0

> **checkIC := normal(subs(t=0,ans),expanded);**

$$checkIC := x(0) = \alpha$$

> **quit**